Lesson No. 20

LES and LDS Instructions

Since the string instructions need their source and destination in the form of a segment offset pair, there are two special instructions that load a segment register and a general purpose register from two consecutive memory locations. LES loads ES while LDS loads DS. Both these instructions have two parameters, one is the general purpose register to be loaded and the other is the memory location from which to load these registers. The major application of these instructions is when a subroutine receives a segment offset pair as an argument and the pair is to be loaded in a segment and an offset register. According to Intel rules of significance the word at higher address is loaded in the segment register while the word at lower address is loaded in the offset register. As parameters segment should be pushed first so that it ends up at a higher address and the offset should be pushed afterwards. When loading the lower address will be given. For example “lds si, [bp+4]” will load SI from BP+4 and DS from BP+6.

1.1. LES and LDS Example

We modify the string length calculation subroutine to take the segment and offset of the string and use the LES instruction to load that segment offset pair in ES and DI.

	; hello world printing with length calculation subroutine

[org 0x0100]

              jmp  start

message:      db   'hello world', 0    ; null terminated string 

;;;;; COPY LINES 005-024 FROM EXAMPLE 7.1 (clrscr) ;;;;;

; subroutine to calculate the length of a string

; takes the segment and offset of a string as parameters

strlen:       push bp

              mov  bp,sp

              push es

              push cx

              push di

              les  di, [bp+4]         ; point es:di to string

              mov  cx, 0xffff         ; load maximum number in cx

              xor  al, al             ; load a zero in al

              repne scasb             ; find zero in the string

              mov  ax, 0xffff         ; load maximum number in ax

              sub  ax, cx             ; find change in cx

              dec  ax                 ; exclude null from length

              pop  di

              pop  cx

              pop  es

              pop  bp

              ret  4

; subroutine to print a string 

; takes the x position, y position, attribute, and address of a null 

; terminated string as parameters

printstr:     push bp

              mov  bp, sp

              push es

              push ax

              push cx

              push si

              push di

              push ds                 ; push segment of string

              mov  ax, [bp+4]  

              push ax                 ; push offset of string

              call strlen             ; calculate string length

              cmp  ax, 0              ; is the string empty

              jz   exit               ; no printing if string is empty

              mov  cx, ax             ; save length in cx

              mov  ax, 0xb800

              mov  es, ax             ; point es to video base

              mov  al, 80             ; load al with columns per row

              mul  byte [bp+8]        ; multiply with y position

              add  ax, [bp+10]        ; add x position

              shl  ax, 1              ; turn into byte offset

              mov  di,ax              ; point di to required location

              mov  si, [bp+4]         ; point si to string

              mov  ah, [bp+6]         ; load attribute in ah

              cld                     ; auto increment mode

nextchar:     lodsb                   ; load next char in al

              stosw                   ; print char/attribute pair

              loop nextchar           ; repeat for the whole string

exit:         pop  di

              pop  si

              pop  cx

              pop  ax

              pop  es

              pop  bp

              ret  8

start:        call clrscr             ; call the clrscr subroutine

              mov  ax, 30        

              push ax                 ; push x position 

              mov  ax, 20

              push ax                 ; push y position 

              mov  ax, 0x71           ; blue on white attribute

              push ax                 ; push attribute 

              mov  ax, message

              push ax                 ; push address of message

              call printstr           ; call the printstr subroutine

              mov  ax, 0x4c00         ; terminate program

              int  0x21

	036
	The LES instruction is used to load the DI register from BP+4 and the ES register from BP+6.

	065
	The convention to return a value from a subroutine is to use the AX register. That is why AX is not saved and restored in the subroutine.


Inside the debugger observe that the segment register is pushed followed by the offset. The higher address FFE6 contains the segment and the lower address FFE4 contains the offset. This is because we have a decrementing stack. Then observe the loading of ES and DI from the stack.

1.2. MOVS Example – Screen Scrolling 

MOVS has the two forms MOVSB and MOVSW. REP allows the instruction to be repeated CX times allowing blocks of memory to be copied. We will perform this copy of the video screen. 

Scrolling is the process when all the lines on the screen move one or more lines towards the top of towards the bottom and the new line that appears on the top or the bottom is cleared. Scrolling is a process on which string movement is naturally applicable. REP with MOVS will utilize the full processor power to do the scrolling in minimum time. 

In this example we want to scroll a variable number of lines given as argument. Therefore we have to calculate the source address, which is 160 times the number of lines to clear. The destination address is 0, which is the top left of the screen. The lines that scroll up are discarded so the source pointer is placed after them. An equal number of lines at the bottom are cleared. These lines have actually been copied above.

	; scroll up the screen

[org 0x0100]

              jmp  start

; subroutine to scroll up the screen

; take the number of lines to scroll as parameter

scrollup:     push bp

              mov  bp,sp

              push ax

              push cx

              push si



              push di

              push es

              push ds

              mov  ax, 80             ; load chars per row in ax

              mul  byte [bp+4]        ; calculate source position

              mov  si, ax             ; load source position in si

              push si                 ; save position for later use

              shl  si, 1              ; convert to byte offset

              mov  cx, 2000           ; number of screen locations

              sub  cx, ax             ; count of words to move

              mov  ax, 0xb800

              mov  es, ax             ; point es to video base

              mov  ds, ax             ; point ds to video base

              xor  di, di             ; point di to top left column

              cld                     ; set auto increment mode

              rep  movsw              ; scroll up

              mov  ax, 0x0720         ; space in normal attribute

              pop  cx                 ; count of positions to clear

              rep  stosw              ; clear the scrolled space

              pop  ds

              pop  es

              pop  di

              pop  si

              pop  cx

              pop  ax

              pop  bp

              ret  2

start:        mov  ax,5

              push ax                 ; push number of lines to scroll

              call scrollup           ; call the scroll up subroutine

              mov  ax, 0x4c00         ; terminate program

              int  0x21


The beauty of this example is that the two memory blocks are overlapping. If the source and destination in the above algorithm are swapped in an expectation to scroll down the result is strange. For example if 5 lines were to scroll down, the top five lines of the screen are repeated on the whole screen. This is where the use of the direction flag comes in.

When the source is five lines below the destination, the first five lines are copied on the first five lines of the destination. However the next five lines to be copied from the source have been destroyed in the process; because they were the same as the first five lines of the destination. The same is the problem with every set of five lines as the source is destroyed during the previous copy. In this situation we must go from bottom of the screen towards the top. Now the last five lines are copied to the last five lines of the destination. The next five lines are copied to next five lines of the destination destroying the last five lines of source; but now these lines are no longer needed and have been previously copied. Therefore the copy will be appropriately done in this case. 

We give an example of scrolling down with this consideration. Now we have to calculate the end of the block instead of the start.

	; scroll down the screen

[org 0x0100]

              jmp  start

; subroutine to scrolls down the screen

; take the number of lines to scroll as parameter

scrolldown:   push bp

              mov  bp,sp

              push ax

              push cx

              push si



              push di

              push es

              push ds

              mov  ax, 80             ; load chars per row in ax

              mul  byte [bp+4]        ; calculate source position

              push ax                 ; save position for later use

              shl  ax, 1              ; convert to byte offset

              mov  si, 3998           ; last location on the screen

              sub  si, ax             ; load source position in si

              mov  cx, 2000           ; number of screen locations

              sub  cx, ax             ; count of words to move

              mov  ax, 0xb800

              mov  es, ax             ; point es to video base

              mov  ds, ax             ; point ds to video base

              mov  di, 3998           ; point di to lower right column

              std                     ; set auto decrement mode

              rep  movsw              ; scroll up

              mov  ax, 0x0720         ; space in normal attribute

              pop  cx                 ; count of positions to clear

              rep  stosw              ; clear the scrolled space

              pop  ds

              pop  es

              pop  di

              pop  si

              pop  cx

              pop  ax

              pop  bp

              ret  2

start:        mov  ax,5

              push ax                 ; push number of lines to scroll

              call scrolldown         ; call scroll down subroutine

              mov  ax, 0x4c00         ; terminate program

              int  0x21


1.3. CMPS Example – String Comparison

For the last string instruction, we take string comparison as an example. The subroutine will take two segment offset pairs containing the address of the two null terminated strings. The subroutine will return 0 if the strings are different and 1 if they are same. The AX register will be used to hold the return value. 

	; comparing null terminated strings

[org 0x0100]

              jmp  start

msg1:         db   'hello world', 0 

msg2:         db   'hello WORLD', 0

msg3:         db   'hello world', 0

;;;;; COPY LINES 028-050 FROM EXAMPLE 7.4 (strlen) ;;;;;

; subroutine to compare two strings

; takes segment and offset pairs of two strings to compare

; returns 1 in ax if they match and 0 other wise

strcmp:       push bp

              mov  bp,sp

              push cx

              push si

              push di

              push es

              push ds

              lds  si, [bp+4]         ; point ds:si to first string

              les  di, [bp+8]         ; point es:di to second string

              push ds                 ; push segment of first string

              push si                 ; push offset of first string

              call strlen             ; calculate string length

              mov  cx, ax             ; save length in cx

              push es                 ; push segment of second string

              push di                 ; push offset of second string

              call strlen             ; calculate string lenth

              cmp  cx, ax             ; compare length of both strings

              jne  exitfalse          ; return 0 if they are unequal

              mov  ax, 1              ; store 1 in ax to be returned

              repe cmpsb              ; compare both strings

              jcxz exitsimple         ; are they successfully compared

exitfalse:    mov  ax, 0              ; store 0 to mark unequal

exitsimple:   pop  ds

              pop  es

              pop  di

              pop  si

              pop  cx

              pop  bp

              ret  8

start:        push ds                 ; push segment of first string

              mov  ax, msg1

              push ax                 ; push offset of first string

              push ds                 ; push segment of second string

              mov  ax, msg2

              push ax                 ; push offset of second string

              call strcmp             ; call strcmp subroutine

              push ds                 ; push segment of first string

              mov  ax, msg1

              push ax                 ; push offset of first string

              push ds                 ; push segment of third string

              mov  ax, msg3

              push ax                 ; push offset of third string

              call strcmp             ; call strcmp subroutine

              mov  ax, 0x4c00         ; terminate program

              int  0x21

	005-007
	Three strings are declared out of which two are equal and one is different.

	044-045
	LDS and LES are used to load the pointers to the two strings in DS:SI and ES:DI.

	070
	Since there are 4 parameters to the subroutine “ret 8” is used.


Inside the debugger we observe that REPE is shown as REP. This is because REP and REPE are represented with the same prefix byte. When used with STOS, LODS, and MOVS it functions as REP and when used with SCAS and CMPS it functions as REPE. 

Exercises

1. Write code to find the byte in AL in the whole megabyte of memory such that each memory location is compared to AL only once.

2. Write a far procedure to reverse an array of 64k words such that the first element becomes the last and the last becomes the first and so on. For example if the first word contained 0102h, this value is swapped with the last word. The next word is swapped with the second last word and so on. The routine will be passed two parameters through the stack; the segment and offset of the first element of the array.

3. Write a near procedure to copy a given area on the screen at the center of the screen without using a temporary array. The routine will be passed top, left, bottom, and right in that order through the stack. The parameters passed will always be within range the height will be odd and the width will be even so that it can be exactly centered. 

4. Write code to find two segments in the whole memory that are exactly the same. In other words find two distinct values which if loaded in ES and DS then for every value of SI [DS:SI]=[ES:SI].

5. Write a function writechar that takes two parameters. The first parameter is the character to write and the second is the address of a memory area containing top, left, bottom, right, current row, current column, normal attribute, and cursor attribute in 8 consecutive bytes. These define a virtual window on the screen.

The function writes the passed character at (current row, current column) using the normal attribute. It then increments current column, If current column goes outside the window, it makes it zero and increments current row. If current row gets out of window, it scrolls the window one line up, and blanks out the new line in the window. In the end, it sets the attribute of the new (current row, current column) to cursor attribute. 

6. Write a function “strcpy” that takes the address of two parameters via stack, the one pushed first is source and the second is the destination. The function should copy the source on the destination including the null character assuming that sufficient space is reserved starting at destination.

